Nothing Ventured, Nothing Gained: Modeling Venture Capital Decisions

This column develops a simple model of a venture capital business to show how *Mathematica* can be used to obtain insight into risky business decisions.

Robert J. Korsan

I have a friend who is finishing a successful career as an engineer in Silicon Valley. She has decided to become a venture capitalist. (Serendipitously, she lives not far from 2700 Sand Hill Road in Menlo Park, California, the location of the largest concentration of computer related venture capitalists in the US and probably the world. This location is sometimes known as "Vulture Gulch.") My friend has many contacts from her previous work in successful startup companies. She also has friends and contacts with money who trust her instincts and who would like to see greater returns than the stock market is likely to produce in the next few years. So, as they say, "to start a business, all you need is a customer."

My friend has asked my help in developing a strategy for investing the funds entrusted to her by her clients. The fundamental question in investment strategy is simple: How much do I invest this year and how much do I hold back to invest next year so that my capital accumulates as fast (and as safely) as possible?

In this column, I will describe some of the rules of thumb that have been learned by venture capitalists and translate those rules into a simple *Mathematica* model. After discussing how one can account for aversion to risk in a business decision, I'll use *Mathematica* to find the fraction of capital that should be invested each year.

As with all models, the initial formulation will be overly simple, so I will discuss how to make the model more realistic.

Some Rules for Venture Capitalists

Most venture capitalists try to invest in new businesses that show a good chance of reaching an initial public offering (IPO). For a venture capitalist, a "good chance" means something like 1 chance in 10. Companies with chances greater than that, say approaching 1 chance in 2, would be able to go to a bank for financing. Experience shows that most startup businesses take three to ten years to get a product out the door, build market share, and develop the reputation necessary for a successful IPO. Venture capitalists usually try to negotiate a stake in the business that will yield about 20 times their investment at the IPO. So, the issue here is balancing the risk of loss against the chance of a large gain in a few years.

To take account of the value of money over time, we translate all future dollars into their present value using a "discount rate." The usual bench mark is US Treasury bills, which provide an essentially riskless way to trade dollars over time. However, as noted above, the appropriate benchmark for our investors is the stock market. The appropriate tradeoff is closer to \$1.00 today versus \$1.09 a year from now, so I can use a present-value factor of `(100/109)^-1` per year.

Before we can put together our *Mathematica* model, we need to look at the question of aversion to risk. Since this may be unfamiliar territory to the reader, I will review utility theory in the next section.

Risk Aversion

Capturing aversion to risk is not a straight-forward proposition. We know that more is better, but 1000 times more money is not 1000 times better. It is possible to have enough of a good thing. Are losing \$1000 and winning \$1000 equally good and bad? Given the choice, would you rather have \$15,868 (for sure) or the chance to play a lottery in which five coins are flipped and I pay you \$16,384 for any outcome except five tails, in which case you pay me \$128? The ambiguity of these questions demonstrates that a clear definition of "aversion to risk" is needed.

The basic ideas of utility theory were first proposed by Bernoulli in 1738. Von Neumann and Morgenstern provided an axiomatization in 1944 that encapsulates the fundamental judgements necessary to describe aversion to risk. This axiomatization is the foundation of utility theory. It treats

Robert J. Korsan is the Managing Director of Decisions, Decisions!, a management consulting firm, and a technical editor of The Mathematica Journal. He received his B.S. in physics and mathematics from Manhattan College, an M.S.E.E. from Carnegie Mellon University, and an M.A. in mathematics from the University of Pittsburgh.

aversion to risk separately from probability judgements. Another approach is to axiomatize probability theory and utility theory jointly. The class of judgements necessary to describe probability and utility are, in fact, very similar. Fundamental work has been done by L. J. Savage to develop such an axiomatization.

The Von Neumann-Morgenstern approach, which I will adopt here, follows from these five axioms:

1. More is preferred to less.
2. There is a "certain equivalent" between the extreme outcomes of a lottery. This axiom says that if I offer you the chance to receive either a Mercedes Benz or a pound of American cheese depending upon the a flip of a coin, you would be willing to accept instead some intermediate sure thing (the "certain equivalent"), such as a Mazda Miata.
3. The certain equivalent can always be substituted for the lottery. In other words, there is no "rush" in taking a chance. You don't care whether you play the lottery or get the equivalent "sure thing."
4. A higher probability of a better prize is preferred.
5. The laws of probability can be used to reduce compound lotteries. For example, if you are offered the choice of 20 flips of a coin for a pound of American cheese or a Mercedes Benz, or a single random choice of one of the 2^{20} outcomes of the 20 coin flips, you would let me decide which lottery is run. Playing the game 20 times is no different from playing the probabilistically equivalent game once. This axiom is sometimes called "no fun in gambling."

If you accept these five axioms, your choices will be determined by a function that translates a dollar amount into "utility." Given a choice between two lotteries, you will apply the utility function to the dollar outcomes of each lottery, take expectations, and choose the lottery with the higher expected utility. Conversely, your selling (or buying) price for a lottery (the "certain equivalent") is the dollar amount that corresponds to its expected utility.

Utility Functions

In theory, utility should be measured relative to one's total wealth for all possible scenarios over all time. In practice, if the range of gains and losses is small compared to our current wealth, we can ignore our wealth and limit the scenarios to those involving the investment only.

Suppose our utility function is u. Suppose also that we have wealth w and that we face a lottery whose outcome is given by a probability distribution o and whose expected value $EV[o]$ is zero. Since the lottery is as likely to decrease our wealth as to increase it, we would pay a small amount p, called the risk premium, to avoid playing this lottery. Utility theory says that the expected value of our utility if we play the lottery should be equal to the utility of the certain event:

$$EV[u(w+o)] = u(w+EV[o]-p) = u(w-p)$$

Expanding the utilities in Taylor series to second order on the left and to first order on the right, we get:

$$EV[u(w) + o\ u'(w) + o^2\,u''(w)/2] = u(w) - p\ u'(w)$$

Since w is not uncertain, we can simplify the left side and solve for p:

$$p = -(v/2)\ u''(w)/u'(w)$$

where $v = EV[o^2]$ is the variance of the lottery. So the approximate risk premium is proportional to the variance. The quantity $-\,u''(w)/u'(w)$ is called the absolute risk aversion (ARA). It measures our risk aversion for a lottery whose outcome is small compared to our current wealth.

The most commonly used utility functions are logarithmic, $u(w) = a + b\log(w-\rho)$, and exponential, $u(w) = a - b\exp(-w/\rho)$. For a particular choice of the parameter ρ, each of these functions defines utility as a "two-point" scale. The units of utility ("utiles") are arbitrary and are determined by choosing values for two points on the scale. (This choice is similar to the way temperature values are assigned to two physical events, such as freezing and boiling water, to define temperature scales, such as Celsius, Fahrenheit, Kelvin, and Rankine.) We can choose any two dollar amounts and assign them arbitrary "utile" values, such as 0 and 100. These values then determine the constants a and b in the utility function.

```
In[1]:= expU[w_] := a1 - b1 Exp[-w/rho];
        logU[w_] := a2 + b2 Log[w + rho];
```

The parameter ρ is sometimes called the "risk-tolerance" or "risk-aversion coefficient" because of its relation to the absolute risk aversion. For the exponential utility function:

```
In[3]:= - expU''[w]/expU'[w]

         1
Out[3]= ---
        rho
```

Thus, the ARA is independent of the size of the gamble. Casual empiricism says that a person's ARA should decrease as they become more wealthy. The logarithmic utility function has this property:

```
In[4]:= -logU''[w]/logU'[w]

            1
Out[4]= -------
        rho + w
```

Let's compare the two utility functions. We will choose the constants a and b so that zero dollars corresponds to zero utility and 1.5 million dollars has a utility of 100. We'll set the ARA of the utility functions equal at zero, so that ρ has the same value for each, say 0.3. The equations for the constants in the utility functions are:

```
In[5]:= eqns =
          {expU[0] == 0, expU[1.5] == 100,
           logU[0] == 0, logU[1.5] == 100}

                                   b1
Out[5]= {a1 - b1 == 0, a1 - -------- == 100,
                                1.5/rho
                               E

          a2 + b2 Log[rho] == 0, a2 + b2 Log[1.5 + rho] == 100}
```

Solving for the constants yields:

```
In[6]:= Solve[eqns /. rho->0.3, {a1, a2, b1, b2}] // First

Out[6]= {a1 -> 100.678, a2 -> 67.195, b1 -> 100.678,
          b2 -> 55.8111}
```

The utility functions are:

```
In[7]:= {expU[w_], logU[w_]} =
             {expU[w], logU[w]} /. % /. rho -> 0.3

                    100.678
Out[7]= {100.678 - ----------, 67.195 + 55.8111 Log[0.3 + w]}
                    3.33333 w
                   E

In[8]:= Plot[{expU[w], logU[w]}, {w, -0.2, 5},
          AxesLabel -> {"$", "utiles"},
          PlotRange -> All,
          PlotStyle -> {GrayLevel[0], GrayLevel[0.5]}]
```



We have chosen the parameters of the two utility functions so that their behavior is similar near zero. The exponential function appears almost flat for large w, so outcomes are regarded as nearly the same. The logarithmic utility function increases more rapidly, so larger risks are tolerated as wealth increases.

Let's look at a simple example of how the utility functions are used in decision making. Suppose I offer you a choice of two lotteries. The first lottery is a 0.75 chance at a pound of American cheese ($\$2\ 10^{-6}$ million) and a 0.25 chance at a Mercedes Benz ($\$60\ 10^{-3}$ million). The second is a 50-50 chance at a Geo Metro ($\$7\ 10^{-3}$ million) and a Mazda Miata ($\$20\ 10^{-3}$ million). Which of these two lotteries do you prefer?

Your preference will be determined by the choice of a utility function. Let's apply both the exponential and logarithmic utility functions. The first lottery has exponential utilities of:

```
In[9]:= Map[expU, {2 10^-6, 60 10^-3}]

Out[9]= {0.000671187, 18.2499}
```

The expected utility is the inner product of the probabilities and these utilities:

```
In[10]:= {0.75, 0.25} . %

Out[10]= 4.56298
```

The second lottery has a lower expected utility:

```
In[11]:= Map[expU, {7 10^-3, 20 10^-3}]

Out[11]= {2.32197, 6.49305}

In[12]:= {0.5, 0.5} . %

Out[12]= 4.40751
```

Since the first lottery has the higher expected utility, you would always choose it. Now let's apply the logarithmic utility function. For the first lottery, the utilities are:

```
In[13]:= Map[logU, {2 10^-6, 60 10^-3}]

Out[13]= {0.000372073, 10.1756}
```

The expected utility is:

```
In[14]:= eu = {0.75, 0.25} . %

Out[14]= 2.54417
```

For the second lottery, we have:

```
In[15]:= Map[logU, {7 10^-3, 20 10^-3}]

Out[15]= {1.2873, 3.60196}

In[16]:= {0.5, 0.5} . %

Out[16]= 2.44463
```

Again, the first lottery has the higher expected utility, so you would always choose it.

Suppose we already owned the right to play the first lottery for the Mercedes. What price would we sell it for? The certain equivalent is the amount whose utility is equal to the expected utility of the lottery. Using the logarithmic utility, we get:

```
In[17]:= FindRoot[eu == logU[ce], {ce, 0}]

Out[17]= {ce -> 0.0139921}
```

Thus, we would sell the right to play the first lottery for $13,992.10.

A Simple Business Model

In this section, I'll use the rules of thumb described above to create a simple business model in *Mathematica*. The model will be parametrized by three quantities: time (`t`), utility (`u`), and current capitalization (`c`). The "value" (utility) of the business in any year, `t`, will be computed by a function `V[t, u, c]`. From this model, I will compute the fraction (`d`) of the current capitalization to invest each year.

When the venture capital business begins, at `t = 0`, the value of the business is simply the utility `u` of the initial capital `c`:

```
In[1]:= V[0, u_, c_] := u[c]
```

At any later time, the venture capitalist decides to invest some fraction (`d`) of her capital in startup companies. One of two outcomes will occur. Either the company will fail, in which case the capital has been lost, or the company succeeds. When the company succeeds, the capitalist recovers the future value of her investment plus the present value of 20 times the investment. Let `PVF` be the present value factor for 5 years (the average number of years to IPO). Let `a = 20*PVF`. The probability that a company succeeds is `p` and the probability that it fails is `(1-p)`. Thus, the value of the investment at time `t` is expectation of the values of the two possible outcomes:

```
In[2]:= V[t_, u_, c_] :=
          p V[t-1, u, (1 + a d) c] + (1-p) V[t-1, u, (1-d) c]
```

Let's investigate the best investment strategy for the exponential utility function. Since the two constants a and b in the utility function are irrelevant to our calculations, I'll set $a = 0$ and $b = -1$. The risk tolerance ρ can be set to `c/5` (this is another useful rule of thumb for many businesses).

```
In[3]:= rho = c/5;
        u1[w_] = -Exp[-w/rho];
```

Thus, our initial capital is worth:

```
In[5]:= V[0, u1, c]

           -5
Out[5]= -E
```

At `t = 1`, we have to maximize the expected utility of our investment:

```
In[6]:= val1 = V[1, u1, c]

            1 - p             p
Out[6]= -(-----------) - ------------
           5 (1 - d)      5 (1 + a d)
          E              E
```

To find the maximum, we differentiate with respect to the decision variable (`d`), set the result equal to zero, and solve for `d`:

```
In[7]:= Solve[D[%, d] == 0, d] // Short[ #, 2]&

        Solve::ifun:
           Warning: Inverse functions are being used by Solve, so
             some solutions may not be found.

Out[7]//Short=
                     1/5  1/5
                    a    p
           Log[-(------------)]
                            1/5
                    (-1 + p)                 Log[<<1>>]
        {{d -> --------------------}, {d -> ----------}, <<2>>,
                      1 + a                    1 + a

                       4/5  1/5  1/5
                   (-1)    a    p
            Log[-(-------------------)]
                                1/5
                        (-1 + p)
          {d -> --------------------------}}
                          1 + a
```

The multiple solutions are due to the fifth roots of negative one. I notice that if we factor –1 out of the denominator of the `Log` term, the second solution will be the one with real values.

```
In[8]:= dRule = %[[2]] /. (-1+p)^e_ -> (-1)^e (1-p)^e

                  1/5  1/5
                 a    p
            Log[-----------]
                          1/5
                   (1 - p)
Out[8]= {d -> ----------------}
                   1 + a
```

Lastly, a numerical decision rule for a given probability `p` of IPO and `a = 20 PVF` is set:

```
In[9]:= aRule = a -> 20 1.09^-5;

In[10]:= ndRule = %% /. aRule

                                           1/5
                                  1.67024 p
Out[10]= {d -> 0.0714356 Log[-------------]}
                                        1/5
                                 (1 - p)
```

A very interesting value of the probability is the one for which the fraction to be invested is zero:

```
In[11]:= Solve[0 == d /. ndRule, p] // First

Out[11]= {p -> 0.0714356}
```

So, as long as we judge that the chance of reaching IPO is greater than about 7%, we will want to invest at least some our money.

Finally, we can see the behavior of the investment fraction as a function of the probability of a successful IPO. The following function calculates the percent invested at `t = 1`:

```
In[12]:= dec1[p_] = 100 d /. ndRule

Out[12]=                           1/5
                          1.67024 p
         7.14356 Log[-------------]
                                1/5
                         (1 - p)

In[13]:= Plot[dec1[p], {p, 0.03, 0.3},
           AxesLabel -> {"Probability", "% Invested"},
           PlotRange -> {0, 2.5}]
```



For the chosen risk tolerance and the optimal investment, the optimum value in the first period will be a constant consisting of two terms:

```
In[14]:= val1Star = val1 /. ndRule

                                 1 - p
Out[14]= -(-------------------------------------------------) -
            5 (1 - 0.0714356 Log[(1.67024 p^(1/5))/(1 - p)^(1/5)])
           E

                               p
          --------------------------------------------------
            5 (1 + 0.0714356 a Log[(1.67024 p^(1/5))/(1 - p)^(1/5)])
           E
```

The expected utility is equal to the utility of the certain equivalent. Thus, the rule used to calculate the certain equivalent from the expected utility is given by:

```
In[15]:= First[Solve[-Exp[-ce/HoldForm[rho]] == Eu1, ce]]

         Solve::ifun:
            Warning: Inverse functions are being used by Solve, so
              some solutions may not be found.

                                  1
Out[15]= {ce -> rho Log[-(---)]}
                                 Eu1
```

Applying this rule to the expected utility at `t = 1` yields:

```
In[16]:= Short[(rho Log[1/Together[-val1Star /. aRule]]) /.
              Log[E^a_/b_]->a-Log[b], 2]

Out[16]//Short=
                                          1/5
                                         p          5.
    (c (5 + <<1>> - Log[12.9986 (-----------)    + p -
                                           1/5
                                    (1 - p)

                    1/5
                   p          5.
        12.9986 (-----------)    p])) / 5
                         1/5
                  (1 - p)
```

I have to force *Mathematica* to reduce the expressions in the second Log term. I also set the capital (`c`) to 1 so the certain equivalent is expressed as a fraction of our initial capital:

```
In[17]:= Short[CE = PowerExpand[%]/. c -> 100, 2]

Out[17]//Short=
    20 (5 + 4.64282 (0.512969 + <<2>>) -

                             1.               2.
                 12.9986 p        12.9986 p
          Log[p + ------------ - ------------])
                           1.              1.
                  (1 - p)         (1 - p)
```

Finally, the certain equivalent of the total investment becomes

```
In[18]:= CEplot1 =
          Plot[Release[CE], {p, 0.07, 0.3},
              AxesLabel -> {"Prob", "C. E."}];
```



This result seems to imply that the returns are not near our targets. However, only a small percentage of our original pool of capital was invested. If the probability of a successful IPO is 0.15, the percent invested is:

```
In[19]:= dec1[0.15]

Out[19]= 1.18618
```

Only about 1% of our capital is invested in the first year. So, the return on the invested capital is the incremental certain equivalent divided by the invested capital multiplied by 100:

```
In[20]:= 100 * (CE - 100 /. p -> 0.15) / %

Out[20]= 49.056
```

Thus, we obtain almost a 50% return on our invested capital. The uninvested portion could be put into the stock market. As time progresses and our capital is becomes fully invested, our overall rate of return will get better and better.

Next, we look at the second year:

```
In[21]:= V[2, u1, c] /. rho -> c/5

                        1 - p                      p
Out[21]= (1 - p) (-(------------) - ---------------------) +
                      5 (1 - d)^2     5 (1 - d) (1 + a d)
                     E               E

                        1 - p                   p
            p (-(---------------------) - --------------)
                   5 (1 - d) (1 + a d)     5 (1 + a d)^2
                  E                        E
```

We can maximize the expected utility as before:

```
In[22]:= (eqn =
  (Numerator[Together[D[%, d] ]]/10 /. aRule ) == 0) //
    Short[#, 2]&

Out[22]//Short=
           5 (1 - d) (1 + 12.9986 d) + 5 (1 + 12.9986 d)^2
        -E                                                 +
            <<11>> == 0
```

The percent invested in the second year (`t = 2`) is determined by the following function:

```
In[23]:= dec2[prob_] :=
           Block[{e = eqn /. p -> prob},
             100 d /. FindRoot[Release[e], {d, 0.02}]]
```

The following plot compares the optimal investment in the first year to the optimal investment in the second year. This varies as our judgement of the chances of reaching IPO increase. For small chances the policy is almost identical. As the probability increases, the second year invests more than the first year, peaking at about $p = 0.27$. Then as the probability increases further, the percent invested once again approaches the investment recommended in the first year.

```
In[24]:= Plot[(dec1[pr]-dec2[pr]), {pr, 0.07, 0.3},
             AxesLabel->{"Prob", "delta % Invested"}];
```



Overall, the recommended investment is essentially identical. The reader is encouraged to work out further periods to see that this is true well into the future, i.e. fix the probability of IPO at about 0.15 and find the recommended investment for the next 10 to 20 years.

Next, we determine the effect of a different utility function.

Logarithmic Utility

Let's investigate the best investment strategy for the logarithmic utility function:

```
In[25]:= u2[c_] = Log[c + rho];
```

Our initial capital is worth:

```
In[26]:= V[0, u2, c]

             6 c
Out[26]= Log[---]
              5
```

I set the risk tolerance to c/5. At t = 1, we have to maximize the expected utility of our investment:

```
In[27]:= rho = c/5;
         val1 = V[1, u2, c]

                     6 c (1 - d)          6 c (1 + a d)
Out[28]= (1 - p) Log[-----------] + p Log[-------------]
                          5                     5
```

We differentiate, set the result equal to zero, and solve for the optimal investment:

```
In[29]:= ndRule = (Solve[D[%, d] == 0, d] // First) /. aRule
Out[29]= {d -> -0.0769312 (1 - 13.9986 p)}
```

The probability at which the fraction to be invested becomes zero is:

```
In[30]:= Solve[0 == d /. ndRule, p] // First
Out[30]= {p -> 0.0714356}
```

Interestingly, it is identical to the result obtained for the exponential utility function. This is because we our utility functions to have identical ARA when our investment is zero.

Now, we can see the behavior of the investment fraction as a function of the probability of a successful IPO. The following function calculates the percent invested at t = 1.

```
In[31]:= dec1[p_] = 100 d /. ndRule;
In[32]:= Plot[dec1[p], {p, 0, 0.3},
             AxesLabel -> {"Prob", "% Invested"},
             PlotRange -> {0, 25}];
```



Our investment decision will be much more aggressive using this utility function.

For the chosen risk tolerance and the optimal investment, the expected utility in the first period will be a constant consisting of two terms.

```
In[33]:= val1Star = val1 /. Append[ndRule, aRule]

                 6 c (1 - 1. (1 - 13.9986 p))
Out[33]= p Log[----------------------------] +
                              5

                     6 c (1 + 0.0769312 (1 - 13.9986 p))
         (1 - p) Log[-----------------------------------]
                                     5
```

The expected utility is equal to the utility of the certain equivalent. Thus, the rule used to calculate the certain equivalent from the expected utility is given by:

```
In[34]:= Solve[Log[ce + rho] == Eu1, ce] // First

                       Eu1
                -c + 5 E
Out[34]= {ce -> -----------}
                     5
```

We set c to 100 to express the certain equivalent as a percent of capital:

```
In[35]:= (Exp[val1Star] - rho /. c->100) /.
           E^(a_ Log[b_] + c_ Log[d_]) -> b^a d^c

Out[35]= -20 + 120 (1 - 1. (1 - 13.9986 p))^p

           (1 + 0.0769312 (1 - 13.9986 p))^(1 - p)

In[36]:= CEplot2 =
           Plot[Release[%], {p, 0.07, 0.3},
             AxesLabel -> {"Prob", "C. E."}];
```



We can compare the effect of the two utility functions:

```
In[37]:= Show[CEplot1, CEplot2];
```



The logarithmic utility function results in much more aggressive behavior and correspondingly greater return. As mentioned previously, it also seems more reasonable since the absolute risk aversion goes down as wealth increases. Once again, only part of our initial capital is invested in the first period, so our returns on invested capital are very large indeed.

The reader is encouraged to work out the investment and returns for period two and higher. What is the relationship of the optimal investment decision from period to period?

More Realistic Models

The only uncertainty we have included in our model is whether or not the IPO will be successful. A more realistic model of the business would treat as uncertain both the length of time to IPO and the return from the IPO. Both of these uncertainties will affect the prescribed behavior of our venture capitalist. The interested reader is invited to make the return uncertain (conditioned) upon reaching an IPO and determine how the recommended investment decision varies as the variance of the uncertain return gets larger.

More realistic models would not really change the overall picture I have drawn here. The recommendations would be a little more conservative since there would be more uncertainty. The next stage would be to introduce various research or marketing milestones that mark the path to IPO. Although these refinements do not provide new insights into the optimal investment strategy, they do allow us to understand the value of gathering information about the business' ability to meet these goals before investing. By explicitly modeling the down-stream decision, we can determine value of the business with and without the added information. If the down-stream decision would change with the new information (hence enhancing the value of the business), then the difference of the two values is the value of the information. Performing such value-of-information computations will be the subject of a future column.

Conclusions

Utility theory lets us make decisions consistent with our aversion to risk. The axioms of utility theory are simple, compelling, and, if we accept them, create a method for behaving consistently across decisions. Simple mathematical models and *Mathematica* tools allow us to apply decision and utility theory easily. Remember always: *insight, not numbers*.

References

Luce, R. Duncan, and Howard Raiffa. 1957. *Games and Decisions*. John Wiley and Sons, Inc. New York.

Feller, William. 1971. *An Introduction to Probability Theory and Its Applications* Volumes 1 and 2. John Wiley and Sons, Inc. New York.

Robert J. Korsan
Decisions, Decisions!, 2136 Lyon Avenue,
Belmont, CA 94002-1639
bobk@decide.com

The electronic supplement contains the notebook Vultures.

